
The latest incarnation of a DES Day Counter was implemented in March 2025.
See: http://darcy.rsgc.on.ca/ACES/Projects/DayCounter/index.html
The code below supports the DES breadboard prototypes developed for the project as an introduction to RF to ICS3U-E. 
Royal St. George’s College		Design Engineering Studio
NRF24L01: DAY COUNTER TRANSMIT & RECEIVE
Advanced Computer Engineering School		ICS3U-E: AVR Foundations

Transmit

// PROJECT  :TxMasterRotaryEncoder
// PURPOSE  :Day Counter Master Transmitter
// DEVICE   :Genuine Nano + nRF24L01 + Rotary Encoder
// AUTHOR   :C. D'Arcy
// DATE     :2025 03 16-...
// uC       :328p (Nano)
// COURSE   :*
// STATUS   :Working.Paired with Large7SegmentReceiver.ino or RxSlave.ino 
//          :It was the knockoff Nano that couldn't handle the power.
//          :When I switched to the Genuine Nano, it worked.
// NOTES    :Can be finicky...
 
#include <SPI.h>
#include <RF24.h>
RF24 radio(9, 10);  //CE (Chip Enable/Disable), CSN(SS)
byte addresses[][6] = { "DESDC", "2Node" };
#define CHANNEL 91  //Any value between 0 and 125 (2.400GHz to 2.525GHz)

#define DEBUG true  //Conditional use of Serial support
#define RE_PINA 2   // A
#define RE_PINB 3   // B
boolean CW = true;  //direction: CW or CCW (not CW)
volatile boolean triggered = false;
uint8_t day = 0;  // self-explanatory

void setup() {
  if (DEBUG) {
    Serial.begin(9600);
    while (!Serial);
  }

  radio.begin();                  //invoke the radio object
  radio.setPALevel(RF24_PA_MIN);  //close range so minimum power sufficient
  radio.setChannel(CHANNEL);      //Tx and Rx communication on same channel
  radio.openWritingPipe(addresses[0]);  //Transmit assumes these pipes
  radio.stopListening();

  attachInterrupt(digitalPinToInterrupt(RE_PINA), ISR_Rotary, RISING);
}




void loop() {
  if (triggered) {
    triggered = false;
    //1. First, determine the direction (hardware (logic) strategy)...
    CW = digitalRead(RE_PINA) ^ digitalRead(RE_PINB);
    //2. Now, determine whether to decrement or increment the day number
    if (CW)
      day = (day + 1) % 10;
    else
      day = (day == 0) ? 9 : day - 1;
    if (DEBUG) {  //Conditional Serial support for debugging purposes
      Serial.println(CW);
      displayStates(day);
    }
    //3. Finally, Transmit...
    radio.write(&day, 1);  //transmit day number, one byte
  }
}

void ISR_Rotary() {  //Keep ISR bodies as short as possible
  triggered = true;
}

//for Debugging purposes, if necessary...
void displayStates(uint8_t dayNumber) {
  Serial.print(dayNumber);
  Serial.print(',');
  uint8_t stateA = digitalRead(RE_PINA);
  uint8_t stateB = digitalRead(RE_PINB);
  Serial.print(stateA);
  Serial.print(',');
  Serial.println(stateB);
}

Receive


// PROJECT  :RxSlave
// PURPOSE  :Day Counter Slave Receiver (for TxMasterRotaryEncoder.ino)
// DEVICE   :Nano + nRF24L01 + 74HC595/6 + Big 7-Segment Display
// AUTHOR   :C. D'Arcy
// DATE     :2025 03 16-...
// uC       :328p (Nano)
// COURSE   :*
// STATUS   :Working (when used with TxMasterRotaryEncoder.ino)
//     :http://darcy.rsgc.on.ca/ACES/TEI3M/CommunicationProtocols.html#RF1
//     :https://tmrh20.github.io/RF24/classRF24.html
//     :http://darcy.rsgc.on.ca/ACES/Datasheets/nRF24L01.pdf
#include <SPI.h>
#include <RF24.h>
#define CHANNEL 91	//any value between 0 and 125 (2.400GHz to 2.525GHz)
uint8_t receivedData; //for receiver to store the incoming byte
RF24 radio(9, 10);    //CE (Chip Enable/Disable), CSN(SS)
uint8_t addresses[][6] = { "DESDC", "2Node" };//pipe names 

#define DEBUG false  //Conditional use of Serial support
#define DATA 7       //74HC595/TPIC6C596 
#define CLOCK 6      //
#define LATCH 5      //
#define RE_PINA 2    // possibly nRF24L01: IRQ at some point
#define RE_PINB 3    // ?

// Developed on Small 0.56" 7-Segment Display
uint8_t segments[] = {
  //abcdefg0
  B11111100, B01100000, B11011010, B11110010,  //0..3
  B01100110, B10110110, B10111110, B11100000,  //4..7
  B11111110, B11100110                         //8..9
};
// Live on Large 6.5" Sparkfun 7-Segment Display (COM-08530)
uint8_t Segments[] = {
  //abcdefg0
  B11111100, B01100000, B11011010, B11110010,  //0..3
  B01100110, B10110110, B10111110, B11100000,  //4..7
  B11111110, B11100110                         //8..9
};
uint8_t numSegments = sizeof(segments);

volatile boolean triggered = false;
uint8_t day = 1;




void setup() {
  if (DEBUG) {
    Serial.begin(9600);
    while (!Serial)
      ;
  }
  //prepare shift register control pins
  pinMode(DATA, OUTPUT);
  pinMode(CLOCK, OUTPUT);
  pinMode(LATCH, OUTPUT);
  // Test shiftout...
  digitalWrite(LATCH, LOW);
  shiftOut(DATA, CLOCK, LSBFIRST, segments[day]);
  digitalWrite(LATCH, HIGH);

  // (Not Implemented yet) External interrupt signalling new RF data...
  attachInterrupt(digitalPinToInterrupt(RE_PINA), ISR_nRFIRQ, CHANGE);

  radio.begin();                  //invoke the radio
  radio.setPALevel(RF24_PA_MIN);  //low power ampl. for close proximity
  radio.setChannel(CHANNEL);      //mutually agreeable frequency 

  radio.openWritingPipe(addresses[1]);	//select ONE addresses to write to
  radio.openReadingPipe(1, addresses[0]);//Rx and Tx are reversed
  radio.startListening();                  //should be good to go...

  if (DEBUG)
    displayAll();
}




void loop() {
  if (radio.available()) {  //anything to be read?
    radio.read(&day, 1);    //grab the day number [0..9]
    digitalWrite(LATCH, LOW);
    shiftOut(DATA, CLOCK, LSBFIRST, segments[day]);
    digitalWrite(LATCH, HIGH);
    if (DEBUG)
      Serial.println(day);  //local echo
  }
  //if and when IRQ interrupt enabled...
  if (triggered) {
    triggered = false;
    digitalWrite(LATCH, LOW);
    shiftOut(DATA, CLOCK, LSBFIRST, segments[day]);
    digitalWrite(LATCH, HIGH);
  }
}

void ISR_nRFIRQ() {  //Keep ISR bodies as short as possible
  triggered = true;
}

//for Debugging purposes if necessary...
void displayAll() {
  for (uint8_t d = 0; d < numSegments; d++) {
    digitalWrite(LATCH, LOW);
    shiftOut(DATA, CLOCK, LSBFIRST, segments[d]);
    digitalWrite(LATCH, HIGH);
    Serial.println(d);
    delay(1000);

}


[image: ]
image1.png
e eeeee
e eeeee
D R ] I R A I I I I I A
D R ] D A I I I I A Y
D R R Y D R R I R R I I
D R ] D I R I I A A )
D I Y I R R R I A
D I I I I I I I R I I I I I
" D I I I I I I I A I I I I A B B A I A A
o000 0 D I A I AR
ce e D I R I A A
'_""'000""000000-oo0ooooo'..........oooooooooooooooooooo
Seeee s eloe eoeee sovee svees eseeee eseees eseees eseees
Se e ee sesee PO PP PP I PP s PP P e PP P e s s e s

fritzing




image2.png
RSGC ACES: Day Counter Prototype - Receiver

Pee e Geeee Geeee eeeee eeeee eeeee
Pe e e Geeee Geeee eeeee eeeee eeeee
° Pe e 000000 o 6 0600600000000 000000000000000000 00
o Pe o000 00 e 0000000000000 0000000000 00
. R e 0000000000000 0000000000000
. X XK - e 0000000000000 0000000000000
e e A o e 0000000000000 0000000000 000
ARDUINO -
. UL o o e 0000000000000 00000000000 00
. o (XXXXXXXe] . e 0000000000000 00000000000 00
aammm— e ¢ e 000000 e . © © 0000000000000 000000000 000
- oy e e e . LA ® © 000000000 0000000000000 0o
DR DR lo v | @906 @©eeee eGeeee weeee eeeee
DR DR @eee ©eeee ©Geeee eeGeee Geeee eGeeee eeeee eGeeee





